

WINDOWS SCRIPT HOST



ENTRADA I SORTIDA · ARGUMENTS · PIPES I REDIRECCIONS

INDEX

ENTRADA I SORTIDA.....	3
ARGUMENTS	5
PIPES I REDIRECCIONS	8
ANNEX 'readStdIn.vbs'	11
ANNEX 'Arguments.vbs'	12
ANNEX 'nomcognoms.vbs'	12
ANNEX 'ucasein.vbs'	13
ANNEX 'llegir_fitxer.vbs'	13

ENTRADA I SORTIDA

L'entrada i sortida per defecte d'un script de Windows són el teclat i el monitor.

El següent script (ReadStdIn.vbs),

```
Option Explicit

'-- -----
'-- Script que llegeix de l'entrada estàndard
'-- i escriu a la sortida estàndard línia a línia
'-- del que va llegint
'-- -----


Dim entrada, sortida, str, i
Set entrada = WScript.StdIn
Set sortida = WScript.Stdout

i = 1

'Llegim de Standard d'entrada fins al final.
Do While Not entrada.AtEndOfStream
    str = entrada.ReadLine
    sortida.WriteLine "Contingut de la línia " & i & " " & str
    i = i + 1
Loop
```

ens mostra per pantalla el que introduïm pel teclat.

Un exemple d'execució (`cscript ReadStdIn.vbs`, finalitzat amb `CTRL+Z`) seria,



```
C:\Windows\system32\cmd.exe

C:\temp>dir *.vbs
El volumen de la unidad C no tiene etiqueta.
El número de serie del volumen es: 223A-724E

Directorio de C:\temp
15/10/2012  16:41                525 readStdIn.vbs
               1 archivos                525 bytes
               0 dirs 223.074.308.096 bytes libres

C:\temp>cscript readStdIn.vbs
Microsoft (R) Windows Script Host versión 5.8
Copyright (C) Microsoft Corporation 1996-2006. Reservados todos los derechos.

hola
Contingut de la línia 1 hola
que tal?
Contingut de la línia 2 que tal?
adeu
Contingut de la línia 3 adeu
^Z

C:\temp>
```

tot el que s'introdueix mitjançant el teclat es mostra per pantalla.

Si es necessari es podem modificar per tal que l'entrada o sortida sigui un fitxer.

ARGUMENTS

Els arguments d'un script de Windows ens permeten indicar des de la línia de comandes com s'ha de comportar un programa.

El següent script (Arguments.vbs),

```
Option Explicit

Dim argsNamed, argsUnnamed, i
Set argsNamed = WScript.Arguments.Named
Set argsUnnamed = WScript.Arguments.Unnamed


Wscript.Echo "Hi ha " & argsUnnamed.Count & " arguments sense nom."
Wscript.Echo "Hi ha " & argsNamed.Count & " arguments amb nom."

For i=0 To argsUnnamed.Count-1
    WScript.Echo "El valor de l'argument amb index " & i & " sense nom és: " & argsUnnamed.Item(i)
Next

If argsNamed.Exists("B") Then
    WScript.Echo "El valor de l'argument /B és " & argsNamed.Item("B")
End If
```

ens indica quins arguments s'han introduït.

Un exemple d'execució (cscript Arguments.vbs a /B:bb ccc /D:dddd eeeee /F:ffffff) seria,



The screenshot shows a Windows command prompt window titled "C:\Windows\system32\cmd.exe". The user has run the command `dir` in the `C:\temp` directory, showing the contents of `Arguments.vbs`. Then, the user has run the command `cscript Arguments.vbs a /B:bb ccc /D:dddd eeeee /F:ffffff`. The script's output is as follows:

```
C:\temp>dir
El volumen de la unidad C no tiene etiqueta.
El número de serie del volumen es: 223A-724E

Directorio de C:\temp
24/10/2013  11:50    <DIR>          .
24/10/2013  11:50    <DIR>          ..
24/10/2013  11:50                520 Arguments.vbs
               1 archivos             520 bytes
               2 dirs  213.640.986.624 bytes libres

C:\temp>cscript Arguments.vbs a /B:bb ccc /D:dddd eeeee /F:ffffff
Microsoft (R) Windows Script Host versión 5.8
Copyright (C) Microsoft Corporation 1996-2006. Reservados todos los derechos.
Hi ha 3 arguments sense nom.
Hi ha 3 arguments amb nom.
El valor de l'argument amb index 0 sense nom és: a
El valor de l'argument amb index 1 sense nom és: ccc
El valor de l'argument amb index 2 sense nom és: eeeee
El valor de l'argument /B és bb
C:\temp>
```

el script mostra els arguments en funció del tipus: sense nom (a, ccc i eeeee) i amb nom (bb, dddd i fffffff).

Un altre exemple seria el següent script (nomcognoms.vbs) que recupera una cadena de text (noms i cognoms d'una persona amb el caràcter "x" enlloc d'espais en ordre invers) a través de dos arguments amb nom (H i T) per després tractar-la i obtenir el nom en un format correcte.

ARGUMENTS	RESULTAT
aicragxz erepxocap	paco perez garcia
soir xoetamxrotceh	hector mateo rios
zerepxaicra gxairam	maria garcia perez

Option Explicit

```

Dim sortida, argsAmbNom
Dim cadenaInversa, nomTemporal, longNomTemporal, posX1, posX2, nomFinal

Set sortida = Wscript.Stdout
Set argsAmbNom = WScript.Arguments.Named

If argsAmbNom.Exists("H") And argsAmbNom.Exists("T") Then
    cadenaInversa = argsAmbNom.Item("H") & argsAmbNom.Item("T")

    nomTemporal = StrReverse(cadenaInversa)
    longNomTemporal = Len(nomTemporal)


    posX1 = InStr(nomTemporal, "x")
    posX2 = InStr(posX1+1, nomTemporal, "x")

    nomFinal = Left(nomTemporal, posX1-1)
    nomFinal = nomFinal & " " & Mid(nomTemporal, posX1+1, posX2-posX1-1)
    nomFinal = nomFinal & " " & Right(nomTemporal, longNomTemporal-posX2)

    sortida.WriteLine nomFinal
Else
    WScript.Echo "Arguments incorrectes!"
End If

```

Exemples de diferents execucions serien:



```
C:\Windows\system32\cmd.exe

C:\Users\admin>cscript nomcognoms.vbs /H:soir /T:xoetamxrotceh
Microsoft (R) Windows Script Host version 5.8
Copyright (C) Microsoft Corporation 1996-2006. Reservados todos los derechos.

hector mateo rios

C:\Users\admin>cscript nomcognoms.vbs /T:xoetamxrotceh /H:soir
Microsoft (R) Windows Script Host version 5.8
Copyright (C) Microsoft Corporation 1996-2006. Reservados todos los derechos.

hector mateo rios

C:\Users\admin>
```

quan s'executa el `script` l'ordre dels arguments amb nom no influeix en el resultat.

PIPES I REDIRECCIONS


Els pipes ens permeten enviar la sortida d'un script o programa a la entrada d'un altre.

```
' ucasein.vbs
' Converteix la entrada estàndard a majúscules i la rediregeix per la sortida

Dim strText

strText = WScript.StdIn.ReadAll

WScript.Stdout.Write Ucase (strText)
```



The screenshot shows a Windows command prompt window titled "C:\Windows\system32\cmd.exe". The user enters the command `dir` in the `C:\temp` directory, which lists the contents including `ucasein.vbs`. Then, the user enters `dir ! cscript ucasein.vbs ! more`, which runs the script and pipes its output to the `more` command. The output of the script is displayed in all caps, showing volume information for drive C and the directory listing for `C:\temp`.

```
C:\Windows\system32\cmd.exe
C:\temp>dir
El volumen de la unidad C no tiene etiqueta.
El número de serie del volumen es: 7C05-F1A1

Directorio de C:\temp
11/12/2014  16:25    <DIR>          .
11/12/2014  16:25    <DIR>          ..
11/12/2014  16:27                184 ucasein.vbs
                1 archivos             184 bytes
                2 dirs   72.955.248.640 bytes libres

C:\temp>dir ! cscript ucasein.vbs ! more
Microsoft (R) Windows Script Host versión 5.8
Copyright (C) Microsoft Corporation 1996-2006. Reservados todos los derechos.

EL VOLUMEN DE LA UNIDAD C NO TIENE ETIQUETA.
EL NÚMERO DE SERIE DEL VOLUMEN ES: 7C05-F1A1

DIRECTORIO DE C:\TEMP
11/12/2014  16:25    <DIR>          .
11/12/2014  16:25    <DIR>          ..
11/12/2014  16:27                184 UCASEIN.UBS
                1 ARCHIVOS             184 BYTES
                2 DIRS   72.955.248.640 BYTES LIBRES

C:\temp>_
```

La sortida del script `ucasein.vbs` es la entrada per la comanda `more`.

També podem redireccionar l'entrada per a que sigui un fitxer, el següent script (llegir_fitxer.vbs),

```
Option Explicit

Sub pEscriu_noms()
    Dim entrada, sortida, nom, contador

    Set entrada = wscript.StdIn
    Set sortida = wscript.Stdout

    contador = 0


    Do while not entrada.AtEndOfStream
        nom = entrada.ReadLine
        contador = contador + 1
        sortida.WriteLine contador & ") " & nom
    Loop

    sortida.WriteLine ""
End Sub

pEscriu_noms
```

mostra per pantalla el que introduïm per l'entrada, en aquest cas la entrada és un fitxer de text.

Un exemple d'execució (cscript llegir_fitxer.vbs < noms.txt) seria,



```
C:\Windows\system32\cmd.exe

C:\temp>dir
El volumen de la unidad C no tiene etiqueta.
El número de serie del volumen es: 223A-724E

Directorio de C:\temp
24/10/2013  09:02    <DIR>          .
24/10/2013  09:02    <DIR>          ..
24/10/2013  09:02                350  llegir_fitxer.vbs
24/10/2013  08:47                33   noms.txt
                2 archivos             383 bytes
                2 dirs  223.072.026.624 bytes libres


C:\temp>cscript llegir_fitxer.vbs < noms.txt
Microsoft (R) Windows Script Host versión 5.8
Copyright (C) Microsoft Corporation 1996-2006. Reservados todos los derechos.

1) paco
2) pepe
3) pepa
4) laura
5) manolo


C:\temp>_
```

mostra per pantalla el contingut del fitxer de text.

El fitxer (noms.txt) conté el següent text,



Una altre manera d'executar-lo seria desviant l'entrada i sortida, de manera que llegiria el fitxer (noms.txt) i el resultat (resultat.txt) el bolcaria a un altre fitxer,



```
C:\Windows\system32\cmd.exe

C:\TEMP>dir
El volumen de la unidad C no tiene etiqueta.
El número de serie del volumen es: 7C05-F1A1

Directorio de C:\TEMP
15/12/2014  22:17    <DIR>          .
15/12/2014  22:17    <DIR>          ..
15/12/2014  22:15                367 llegir_fitxer.vbs
15/12/2014  22:14                33 noms.txt
                2 archivos             400 bytes
                2 dirs  37.366.747.136 bytes libres

C:\TEMP>cscript //nologo llegir_fitxer.vbs < noms.txt > resultat.txt

C:\TEMP>dir
El volumen de la unidad C no tiene etiqueta.
El número de serie del volumen es: 7C05-F1A1

Directorio de C:\TEMP
15/12/2014  22:18    <DIR>          .
15/12/2014  22:18    <DIR>          ..
15/12/2014  22:15                367 llegir_fitxer.vbs
15/12/2014  22:14                33 noms.txt
15/12/2014  22:18                50 resultat.txt
                3 archivos             450 bytes
                2 dirs  37.366.747.136 bytes libres

C:\TEMP>type resultat.txt
1) paco
2) pepe
3) pepa
4) laura
5) manolo

C:\TEMP>
```

ANNEX 'readStdIn.vbs'

Option Explicit

```
'-- -----  
'-- Script que llegeix per l'entrada estàndard  
'-- i escriu a la sortida estàndard línia a línia  
'-- del que va llegint  
'-- -----  
  
Dim entrada, sortida, str, i  
Set entrada = WScript.StdIn  
Set sortida = WScript.Stdout  
  
i = 1  
  
'Llegim de Standard d'entrada fins al final.  
Do While Not entrada.AtEndOfStream  
    str = entrada.ReadLine  
    sortida.WriteLine "Contingut de la línia " & i & " " & str  
    i = i + 1  
Loop  
  
WScript.Quit
```

ANNEX 'Arguments.vbs'

```
Option Explicit

Dim argsNamed, argsUnnamed, i
Set argsNamed = WScript.Arguments.Named
Set argsUnnamed = WScript.Arguments.Unnamed

Wscript.Echo "Hi ha " & argsUnnamed.Count & " arguments sense nom."
Wscript.Echo "Hi ha " & argsNamed.Count & " arguments amb nom."

For i=0 To argsUnnamed.Count-1
    WScript.Echo "El valor de l'argument amb index " & i & " sense nom és: " & argsUnnamed.Item(i)
Next

If argsNamed.Exists("B") Then
    WScript.Echo "El valor de l'argument /B és " & argsNamed.Item("B")
End If

WScript.Quit
```

ANNEX 'nomcognoms.vbs'

```
Option Explicit

Dim sortida, argsAmbNom
Dim cadenaInversa, nomTemporal, longNomTemporal, posX1, posX2, nomFinal

Set sortida = Wscript.StdOut
Set argsAmbNom = WScript.Arguments.Named

If argsAmbNom.Exists("H") And argsAmbNom.Exists("T") Then
    cadenaInversa = argsAmbNom.Item("H") & argsAmbNom.Item("T")

    nomTemporal = StrReverse(cadenaInversa)
    longNomTemporal = Len(nomTemporal)

    posX1 = InStr(nomTemporal, "x")
    posX2 = InStr(posX1+1, nomTemporal, "x")

    nomFinal = Left(nomTemporal, posX1-1)
    nomFinal = nomFinal & " " & Mid(nomTemporal, posX1+1, posX2-posX1-1)
    nomFinal = nomFinal & " " & Right(nomTemporal, longNomTemporal-posX2)

    sortida.WriteLine nomFinal
Else
    WScript.Echo "Arguments incorrectes!"
End If

WScript.Quit
```

ANNEX 'ucasein.vbs'

```
' ucasein.vbs
' Converteix l'entrada estàndard a majúscules i la rediregeix per la sortida

Dim strText

strText = WScript.StdIn.ReadAll

WScript.Stdout.Write Ucase (strText)

WScript.Quit
```

ANNEX 'llegir_fitxer.vbs'

```
Option Explicit

Sub pEscriu_noms()
    Dim entrada, sortida, nom, contador

    Set entrada = wscript.StdIn
    Set sortida = wscript.Stdout

    contador = 0

    Do while not entrada.AtEndOfStream
        nom = entrada.ReadLine
        contador = contador + 1
        sortida.WriteLine contador & ") " & nom
    Loop

    sortida.WriteLine ""
End Sub

pEscriu_noms

WScript.Quit
```